# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=unicode-emoji
pkgver=15.0
pkgrel=1
pkgdesc="Unicode Emoji Data Files"
arch=(any)
license=(BSD)
url="https://www.unicode.org/emoji/"
depends=('unicode-character-database')
source=("emoji-sequences-$pkgver.txt::http://www.unicode.org/Public/emoji/$pkgver/emoji-sequences.txt"
        "emoji-test-$pkgver.txt::http://www.unicode.org/Public/emoji/$pkgver/emoji-test.txt"
        "emoji-zwj-sequences-$pkgver.txt::http://www.unicode.org/Public/emoji/$pkgver/emoji-zwj-sequences.txt"
        LICENSE)
sha512sums=('8081b75cee79bbeea6c01f6bfba207ea03fbc72226a093057813c65ccfbcbf5276bbbca6385527ab39ea2fa8c39f4d4c3e086889caf2eb4d49805ebee531f632'
            '894d2036867f2a117b8e5fd59b33148b3bb88d09b022d4a09f57df7e6135c3468b9c3d0fa8c80010ace3b916190b8039ad622f04b790abb1db8697c3f2050015'
            '84cd29d74d5e10a364e5ca495f63aa310b0741d54ae483d6dbb8d31b8397f5d55a69e66ab95e655af6bb5f9693164de6b909172ee8e96b4d276aff59977c85a4'
            '9ba78d68ed39745e55538180f132ff47423edb46adde7ac3e62a489870d153808bcf2db57b77e6a7cdae9b4fa03aba99cf228315b9a15be090bedef00b9e69ed')

package() {
  for _f in sequences test zwj-sequences; do
    install -Dm644 emoji-$_f-$pkgver.txt "$pkgdir/usr/share/unicode/emoji/emoji-$_f.txt"
  done

  install -Dm644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
